

Uge05 SD E17, opgave 5

Løs ligningssystemet:

$$\left. \begin{aligned} a \cdot x_1 + x_2 + x_3 &= 1 \\ x_1 + a \cdot x_2 + x_3 &= 1 \\ x_1 + x_2 + a \cdot x_3 &= 1 \end{aligned} \right\}$$

Opgaven er helt central.

Den er svær og forudsætter både teori fra matematikken og erfaring med Maple.

VIGTIGT: Når der indgår en parameter (her a) i ligningssystemet, skal man være meget omhyggelig ved løsningen.

Maples rutiner "LinearSolve" og "ReducedRowEchelonForm" tager ikke hensyn til, at opgaven kan få et andet udfald i visse specielle tilfælde!

Det er overladt til opgaveløseren.

> restart

> with(LinearAlgebra) :

Koefficientmatricen opskrives:

$$> A := \begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix}$$

$$A := \begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix} \quad (1)$$

Højresiden opskrives på matrixform:

$$> b := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$b := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \quad (2)$$

Ligningssystemets totalmatrice opskrives:

> T := <A|b>

$$T := \begin{bmatrix} a & 1 & 1 & 1 \\ 1 & a & 1 & 1 \\ 1 & 1 & a & 1 \end{bmatrix} \quad (3)$$

Metode 1: Løsning ved brug af *ReducedRowEchelonForm*

Først laves 3 rækkeoperationer, som er gyldige for alle a :

- > $T1 := \text{RowOperation}(T, [1, 3]);$
 $T2 := \text{RowOperation}(T1, [2, 1], -1);$
 $T3 := \text{RowOperation}(T2, [3, 1], -a)$

$$T1 := \begin{bmatrix} 1 & 1 & a & 1 \\ 1 & a & 1 & 1 \\ a & 1 & 1 & 1 \end{bmatrix}$$

$$T2 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & a-1 & 1-a & 0 \\ a & 1 & 1 & 1 \end{bmatrix}$$

$$T3 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & a-1 & 1-a & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix} \quad (1.1)$$

Nu er 1. søjle gjort færdig. Et initial-et-tal i 1. række og 0'er nedenunder.

Da vi nu vil dividere med $a - 1$, må vi forudsætte at dette tal ikke er 0!

Antag: $a \neq 1$:

- > $T4 := \text{RowOperation}(T3, 2, \frac{1}{a-1});$
 $T4 := \text{simplify}(T4);$
 $T5 := \text{RowOperation}(T4, [1, 2], -1);$
 $T6 := \text{RowOperation}(T5, [3, 2], a-1)$

$$T4 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & 1 & \frac{1-a}{a-1} & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T4 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T5 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T6 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & -a^2-a+2 & 1-a \end{bmatrix} \quad (1.2)$$

Nu er 2. søjle gjort færdig. Et initial-et-tal i 2. række og 0 ovenover og nedenunder.

Matrix-elementet $-a^2 - a + 2$ på plads 3,3 vil vi gerne ændre til 1 ved en rækkeoperation. Men det forudsætter, at tallet ikke er 0!

$$\begin{aligned} > \text{solve}(-a^2 - a + 2 = 0, a) \\ & \qquad \qquad \qquad -2, 1 \end{aligned} \qquad (1.3)$$

NB: Matrixelementet kan udtrækkes som $T6[3, 3]$:

$$\begin{aligned} > \text{solve}(T6[3, 3] = 0, a) \\ & \qquad \qquad \qquad -2, 1 \end{aligned} \qquad (1.4)$$

Vi ser altså, at vi ikke kan komme videre umiddelbart med $a = -2$. Vi har bagefter 2 specieltilfælde at undersøge: $a = 1$ og $a = -2$.

Antag: $a \neq 1$ og $a \neq -2$:

$$> T7 := \text{RowOperation}\left(T6, 3, \frac{1}{2 - a^2 - a}\right);$$

$$T7 := \text{simplify}(T7);$$

$$T8 := \text{RowOperation}(T7, [2, 3], 1);$$

$$T9 := \text{RowOperation}(T8, [1, 3], -1 - a);$$

$$T9 := \text{simplify}(T9)$$

$$T7 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & \frac{1-a}{-a^2-a+2} \end{bmatrix}$$

$$T7 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & \frac{1}{a+2} \end{bmatrix}$$

$$T8 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & 0 & \frac{1}{a+2} \\ 0 & 0 & 1 & \frac{1}{a+2} \end{bmatrix}$$

$$T9 := \begin{bmatrix} 1 & 0 & 0 & 1 + \frac{-1-a}{a+2} \\ 0 & 1 & 0 & \frac{1}{a+2} \\ 0 & 0 & 1 & \frac{1}{a+2} \end{bmatrix}$$

(1.5)

$$T9 := \begin{bmatrix} 1 & 0 & 0 & \frac{1}{a+2} \\ 0 & 1 & 0 & \frac{1}{a+2} \\ 0 & 0 & 1 & \frac{1}{a+2} \end{bmatrix} \quad (1.5)$$

Delkonklusion:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$

Antag: $a = 1$:

> $\text{subs}(a = 1, T)$

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 \end{bmatrix} \quad (1.6)$$

Nu indgår ingen parameter a , så er det ufarligt at anvende "ReducedRowOperationForm":

> $\text{ReducedRowEchelonForm}(\text{subs}(a = 1, T))$

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (1.7)$$

Delkonklusion:

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Antag: $a = -2$:

> $\text{subs}(a = -2, T)$

$$\begin{bmatrix} -2 & 1 & 1 & 1 \\ 1 & -2 & 1 & 1 \\ 1 & 1 & -2 & 1 \end{bmatrix} \quad (1.8)$$

Nu indgår ingen parameter a , så er det ufarligt at anvende "ReducedRowOperationForm":

> $\text{ReducedRowEchelonForm}(\text{subs}(a=-2, T))$

$$\begin{bmatrix} 1 & 0 & -1 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(1.9)

Delkonklusion:

Hvis $a = -2$:

Ingen løsning.

Samlet konklusion:

Den fuldstændige løsning til ligningssystemet er:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$

Hvis $a = -2$:

Ingen løsning.

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Metode 2: Når vi har lært om begrebet *determinant*, så løses opgaven lettere

> $\text{Determinant}(A)$

$$a^3 - 3a + 2 \quad (2.1)$$

> $\text{solve}(\text{Determinant}(A) = 0, a)$

$$-2, 1, 1 \quad (2.2)$$

Det betyder, at rangen af A er 3, når $a \in \mathbb{R} \setminus \{-2, 1\}$.
Og at rangen er < 3 , når $a = -2$ eller $a = 1$.

Den videre løsning må så opdeles i 3 tilfælde!

Antag, at $a \in \mathbb{R} \setminus \{-2, 1\}$.

"LinearSolve" kan anvendes uden problemer, da matricen A er regulær (har determinant $\neq 0$).

> $\text{LinearSolve}(A, b)$

$$\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$$

(2.3)

Dvs. netop én løsning.

Antag nu, at $a = -2$.

Så kan systemet igen løses let, idet vi fortæller Maple, at $a = -2$:

> $\text{LinearSolve}(\text{subs}(a = -2, A), b)$

Error, (in LinearAlgebra:-BackwardSubstitute) inconsistent system

Dvs. ingen løsning!

Antag nu, at $a = 1$.

Så kan systemet igen løses let, idet vi fortæller Maple, at $a = 1$:

> $\text{LinearSolve}(\text{subs}(a = 1, A), b)$

$$\begin{bmatrix} 1 - t_{2,1} - t_{1,1} \\ -t_{2,1} \\ -t_{1,1} \end{bmatrix}$$

(2.4)

Dvs. uendelig mange løsninger, givet ved 2 frie parametre.

Konklusion:

Den fuldstændige løsning til ligningssystemet er:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$

Hvis $a = -2$:

Ingen løsning.

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Lad os undersøge rangen:

NB: Maple skelner ikke mellem værdierne af a , som er specielle! Maple påstår her, at rangen altid er 3.

> $\text{Rank}(A);$
 $\text{Rank}(T)$

3

3

(2.5)

> $\text{Rank}(\text{subs}(a=-2, A));$
 $\text{Rank}(\text{subs}(a=-2, T))$

2

3

(2.6)

> $\text{Rank}(\text{subs}(a=1, A));$
 $\text{Rank}(\text{subs}(a=1, T))$

1

1

(2.7)

Der er altså løsninger, når blot $a \neq -2$, idet $\text{rang}(A) = \text{rang}(T)$.

Antal variabel $n = 3$.

Antal frie variable = $n - \text{rang}(A)$.

Idet generelle tilfælde er $n - \text{rang}(A) = 3 - 3 = 0$, så der er kun den ene løsning.

Når $a = 1$ er $n - \text{rang}(A) = 3 - 1 = 2$. Løsningen er således 2-dimensionel.